

はじめてのPICマイコン入門〈第7回〉

EEPROMの読み書きを覚えて曲の録音/再生をしてみよう!

録音機能付き電子ピアノの製作
(後編)

落合 正弘
Masahiro Ochiai



電子ピアノ(写真7-1)を製作します。キーボードと同じようにスイッチを並べ、プログラムによって音を鳴らします。次に鳴らした曲をPICマイコンに記憶させ、何回も再生する録音機能を追加します。

スイッチの状態を検出し音を鳴らすプログラム

● メイン・プログラム

リスト7-1に示します。録音再生スイッチが押されていたら録音ルーチンにジャンプします。①の部分です。次に②で音階設定スイッチをチェックして、プリスケアラの分周比を設定します。③でgettonecodeを呼び出して、今どのキーが押されているか、または押されていないかをチェックします。idx変数に結果が返りますのでその値を判断し、キーが押されていた場合はdivconstに分周比を設定し、割り込みを許可します。そうでない場合は割り込みを禁止にして音が出ないようにします。

● タイマ割り込み

割り込み処理ルーチンで出力ポートを反転させます。リスト7-2の①の部分で行っています。

その後、②のようにdivconstの値をTMR0に設定し、次回の割り込み周期を設定しています。

● 任意の周波数を発生させる方法

音を鳴らすためにタイマ0を使って割り込みを発生させます。音階の周波数を出すには割り込み発生周期を音階に応じて変える必要があります。周期を変える際、分周比をいくつにすればよいかを考えます。

割り込み周期を変えるための要素は二つあります。プリスケアラに設定する分周比と、カウント・アップするTMR0に書き込む初期値(図7-1)です。ここでドの音は261.63 Hzです。261.63 Hzの音を出すには、その倍の523.26 Hzで割り込みをしなければなりません。図7-2のように波形の山と谷の1セットが1周期なので、山と谷を変化させる回数は発生周波数の倍が必要




写真7-1
製作した録音/再生機能付き電子ピアノの外観




図7-1 タイマによる割り込みの設計

リスト7-1 メイン・プログラム

```

;
;      メインプログラム
;
main
    BTFSS    PORTA, RECPLAY          ; REC/PLAYスイッチが押されたら...
    goto    record                   ; recordにジャンプ      ①

    MOVLW   01000101b               ; FOSCを64分周
    BTFSS   PORTA, TONESW
    MOVLW   01000100b               ; FOSCを32分周 (1オクターブ上) } ②

    BSF     STATUS, RP0              ; バンク1を選択
    MOVWF   OPTION_REG
    BCF     STATUS, RP0              ; バンク0に戻す

    CALL    gettonecode              ; ボタンコードを取得する

    BTFSS   idx, 7                   ; ボタンが押されていた?
    goto    main_2                   ; 押されていない場合はmain_2へジャンプ

    MOVF    idx, W
    ANDLW   0x0f
    CALL    gettone                  ; 音程用分周定数を取得する
    MOVWF   divconst
    BSF     INTCON, T0IE              ; タイマー0割り込みを許可する
    GOTO    main

main_2
    BCF     INTCON, T0IE              ; タイマー0割り込みを禁止する
    GOTO    main

```

リスト7-2 タイマ割り込み

```

    ORG     0x004                    ; リセットベクタ
    ; (割り込み処理は必ずここにジャンプする)
;-----+ w, STATUSレジスタの退避処理
    MOVWF   w_temp                    ; wレジスタを対比する
    SWAPF   STATUS, W                 ; STATUSレジスタをwレジスタに移動
    MOVWF   status_temp               ; その値をstatus_tempにコピー
;-----+

    CLRF    STATUS

    MOVF    divconst, W                ; 周期を195にするために61 (256-195=61)を書き込む } ②
    MOVWF   TMR0                       ; TMR0に書き込む

    MOVLW   1<<SPEAKER                ; 出力をビット反転する } ①
    XORWF   PORTA, F
;-----+

    BCF     INTCON, T0IF              ; TIMER0の割り込み処理が終わったらT0IFをクリアする必要がある。
    ; これを忘れるとRETFIEの後、すぐにまた割り込みが発生する

;-----+ w, STATUSレジスタの復帰処理
    SWAPF   status_temp, W
    MOVWF   STATUS                     ; status_tempに保存していたSTATUSレジスタを復帰
    SWAPF   w_temp, F
    SWAPF   w_temp, W
;-----+

    RETFIE                             ; 割り込み処理の終了はRETFIE

```